- La function LMV_TS_1_pv realizza il metodo di EULERO ESPLICITO a passo variabile (vedere pag. successiva).

- Esperimenti:

  Pb di Cauchy...
$$\begin{cases} \dot{x} = -x + \operatorname{sen} t \\ x(0) = 4 \end{cases} \quad \text{su } [0,10]$$

  (soluzione esatta:
$$\tilde{x}(t) = \frac{9}{2} e^{-t} + \frac{\operatorname{sen} t - \cos t}{2}$$
)

Si utilizza la proc LMV_TS_1_pv per det approssimazioni della soluzione $\tilde{x}(t)$, con valori di EL_MAX (errore locale massimo richiesto) decrescenti.

Risultati:

con EL_MAX = ...	si ottiene ET_{max} = ...	in N = ... passi
10^{-3}	$2{,}4 \; 10^{-2}$	164
10^{-4}	$7{,}4 \; 10^{-3}$	517
10^{-5}	$2{,}4 \; 10^{-3}$	1643

```
function [T, X, PASSO, StimaEL] = LMV_TS_1_pv(x0, t0, tf, fct, fct2, EL_MAX, dialogo)
//
//  Integra numericamente, sull'intervallo [t0,tf], il problema
//  di Cauchy in R(n):
//  .
//  x = F(t,x)
//  x(t0) = x0
//
//  con il metodo TS(1) - Eulero esplicito - a passo variabile, senza iterazione
//  per la scelta del passo.
//
//  x0: condizione iniziale (colonna di n elementi)
//  t0: istante iniziale
//  tf: istante finale
//  fct: function per F - fct(t,x) deve essere una colonna
//  fct2: function il cui valore fct2(t,x) è la derivata seconda in t della
//        soluzione dell'equazione differenziale che all'istante t assume valore x.
//  EL_MAX: errore locale massimo consentito
//  dialogo: se "loquace" mostra gli istanti di integrazione e, ad ogni iterazione,
//           se è stato necessario aumentare o diminuire la lunghezza del passo
//
//  T = [T(1),...,T(N)], nodi
//  X: matrice n x N - la colonna X(:,i) è la soluzione numerica
//     all'istante T(i)
//  PASSO: riga con PASSO(k) = h tale che T(k+1) = T(k) + h
//  StimaEL: riga delle stime dell'errore locale
//
//
//
//
n = length(x0); // determina il numero di equazioni del sistema
h_min = (tf - t0)/1d6; // passo minimo consentito
T = [];
X = [];
PASSO = [];
StimaEL = [];
//
T(1,1) = t0;
X(:,1) = x0;
StimaEL(1,1) = 0;
//
// ciclo principale
//
while (T(1,$) < tf) & (PASSO(1,$) > h_min | PASSO(1,$) == []),
         // l'iterazione si arresta se si è raggiunto tf o se il passo
         // necessario per ottenere StimaEL = EL_MAX è inferiore a h_min
    // passo massimo per questa iterazione:
    h_max_loc = tf - T(1,$);
    // determina passo
    Nd2x = norm(fct2(T(1,$),X(:,$)));
    if Nd2x == 0 then
       if PASSO == [] then PASSO(1,$+1) = min(h_min * 100, h_max_loc);
       else PASSO(1,$+1) = min(PASSO(1,$), h_max_loc); end;
    else PASSO(1,$+1) = min(sqrt(2*EL_MAX/Nd2x), h_max_loc);
                 // passo per avere StimaEL = EL_MAX (o non superare tf)
    end;
    // calcola nuovo X e T
    X(:,$+1) = X(:,$) + PASSO(1,$)*fct(T(1,$),X(:,$));
    T(1,$+1) = T(1,$) + PASSO(1,$);
    StimaEL(1,$+1) = (1/2) * Nd2x * PASSO(1,$)^2;
    if dialogo == "loquace" then printf("\nT = %3.2e",T($)); end;
end;
if T(1,$) < tf then
   printf("\n\nIntegrazione interrotta a T = %3.2e", T(1,$));
   printf("\n\nh_min = %3.2e , h = %3.2e\n", h_min, PASSO(1,$));
end;
if dialogo == "loquace" then printf("\n"); end;
//
```

Per EL_MAX $= 10^{-4}$ si ottengono, inoltre, i grafici riportati nella pag seguente.

- Come possiamo aspettarci che cambino ET_{max} ed N al decrescere di EL_MAX :

si ricordi che $EL_k \simeq \frac{1}{2} |z''(t_{k-1})| h_{k-1}^2$

$$\Rightarrow \boxed{EL \equiv h^2}$$

$$\Rightarrow h_k = \sqrt{\frac{2\ \text{EL\_MAX}}{|z''(t_k)|}} \geq \sqrt{\frac{2\ \text{EL\_MAX}}{M}} = h_{min}$$

$$|z''(t_k)| \leq M$$

$$\Rightarrow \boxed{h_{min} \equiv \sqrt{\text{EL\_MAX}}}$$

$$\Rightarrow N_{max} = \frac{t_f - t_0}{h_{min}} = (t_f - t_0)\sqrt{\frac{M}{2\ \text{EL\_MAX}}}$$

$$\Rightarrow \boxed{N_{max} \equiv \frac{1}{\sqrt{\text{EL\_MAX}}}}$$

Infine: $\forall k,\ ET_k \leq \underbrace{\text{EL\_MAX}\ N_{max}\ e^{L(t_f - t_0)}}$

$$\boxed{ET_M \equiv \sqrt{\text{EL\_MAX}}}$$



Problema: dx/dt = -Ax + g

Procedura: LMV_TS_1_pv

SUP(k) = EL_MAX + SUP(k-1)*exp(-PASSO(k-1))

EL_MAX = 1.0D-04

Errore totale massimo = 7.4D-03

Numero passi = 517

• Nell'esperimento:

$$EL\_MAX \longleftarrow 10^{-1}\ EL\_MAX$$

$$\Rightarrow \quad h_{min} \longleftarrow h_{min}/\sqrt{10}$$

$$N_{max} \longleftarrow N_{max}\sqrt{10}$$

$$ET_M \longleftarrow ET_M/\sqrt{10}$$

I valori ottenuti sono coerenti con queste previsioni. In particolare:

1. Per EL_MAX = 10^{-5} il numero di passi è ragguardevole!

2. Per ridurre l'errore totale di un fattore 10 occorre ridurre EL_MAX di un fattore 100 e incrementare di un fattore 10 il numero di passi.

$\boxed{\text{Pb: come ridurre l'err totale in modo meno oneroso?}}$

- Nel metodo di EULERO ESPLICITO si ha $EL \equiv h^2$. Questo si esprime dicendo che "il metodo di E. esplicito ha ORDINE UNO"

$$EL \equiv h^2, \quad ET_M \equiv \sqrt{EL} \equiv h^{1}$$

- Se $EL \equiv h^3$ si avrebbe:

$$h_{min} \equiv \sqrt[3]{EL\_MAX}$$

$$N_{max} \equiv \frac{1}{\sqrt[3]{EL\_MAX}}$$

$$ET_M \equiv (EL\_MAX)^{1-\frac{1}{3}} = (EL\_MAX)^{\frac{2}{3}}$$

($EL \equiv h^3 \Rightarrow ET_M \equiv h^2$, metodo di ORDINE DUE; si ottiene, ad es utilizz lo sv di Taylor di $z(t)$ arrestato al 2° ordine ...)

Es:

	$EL \equiv h^2$		$EL \equiv h^3$
	10^{-2}	EL	10^{-2}
	10^{-1}	h_{min}	$\sqrt[3]{10^{-2}} \simeq 2{,}2\ 10^{-1}$
	10	N_{max}	$\sqrt[3]{10^{2}} \simeq 4{,}6$
	10^{-1}	ET_M	$10^{-4/3} \simeq 4{,}6\ 10^{-2}$